--Chapter One--

                            Introduction

        Welcome to Basic Basic, a grouping of lessons on how to use

   the QBasic programming language.  If you are reading this, then it

   is most likely that you want to learn how to program in QBasic, and

   have several questions.  I hope to answer all of them and more in

   this book.

        First, you may be wondering what QBasic exactly is.  Well, there

   is a simple answer - QBasic is a programming language written for

   computers back in 1975, by Bill Gates and Paul Allen, and has been

   the accepted standard ever since.  Why?  Because of its ease of use,

   its English-like commands, and its power.

        QBasic stands for Beginner's All-Purpose Symbolic Instruction

   Code.  Several of its commands are pure English - PRINT, LET, and

   many others.  It has a simple structure for its programs: its lines

   are numbered(10, 20, 30, etc.) and are executed in order.

        But why should you use QBasic?  What's in it for you?  Many

   things are in it for you.  When you learn QBasic, you also learn many

   of the fundamentals of other programming languages.  You also

   can create programs easily.  Once you get into it, you'll find that

   the fun in creating programs is worth coming back to.

        By the end of this book, you should know many things about the

   QBasic programming language, and you will be able to write excellent

   programs that are useful to you and your friends.

                        --Using QBasic--

        QBasic is very easy to use.  Almost every single computer has

   it included with it.  If you don't already know where QBasic is

   located on your computer, check the \DOS directory, or use a program

   to find QBASIC.EXE.

        To start using QBasic, load it up from the DOS prompt, and it

   should look something like the DOS editor. I know you'd like to get

   started right away, so I'll cut to the chase.  Here's a sample program

   for you to type in:

--begin program---------------

CLS

PRINT"Hello!"

PRINT

PRINT"This is my first program!"

--end program----------------

        This is just a list of commands that the computer will interpret

    and execute.  Go up to the "RUN" menu and click "START".   The scren

    will clear, and look something like this, with a "Press any key to

    continue" on the bottom of the screen.

----------------------------

Hello!

This is my first program!

----------------------------

        Notice it doesn't say CLS, or PRINT, or anything.  It does give you

   a "Press any key to continue" message, but that's just QBasic's way of

   telling you it's done with it's work.

        To begin with the explanation, the command 'CLS' stands for CLear

   Screen, and PRINT is quite self-explanatory.  Just make sure that when

   you're PRINTing, have quotes on each side of what you want the screen

   to display.  Fool around with the two commands until you get used to them.

   People who are already familiar with the menus at the top of the screen

   (and their function) can skip the next paragraph.

        To begin a new program, go to the "FILE" menu and click "NEW."

   BEWARE - any program in memory will be LOST FOREVER if it is not saved.

   If you want to save a program, go to the "FILE" menu and click "SAVE."

   QBasic will prompt you for a file name, and your program will be saved

   to disk.  If you want to load that program again, go to the "FILE" menu

   and click "LOAD".  You will get a dialog of what QBasic files are in your

   directory.  Double-click on the one you want to load.

   Questions and Exercises

   1.  Create a program with this screen output:

-begin output--------------

Welcome to Computer Trivia, by Yours Truly.

Did you know that a kilobyte is not just

1000 bytes, but 1024?

-end output----------------

   2.  What is wrong with the following program?

-begin program-------------

CLS

PRINT "Hi

PRINT "This is a cool program by Me!"

-end program---------------

                        --Conclusion--

        You now know the two most QBasic commands and building blocks of

    QBasic programs - CLS and PRINT.  Be ready for more fun ways to learn

    information as well as make programs for yourself, your friends, and

    anyone you want.  Have fun, this is just the beginning...

        Welcome!  In this chapter, we will cover the following commands:

  þ LET

  þ INPUT

        Let's get started.  In the previous chapter, you learned how to

  use the PRINT and CLS commands.  In this chapter, you will learn about

  variables and the INPUT command.

        What are variables?  Variables are "boxes" in the computer's memory

  to store a value - be it a number, name, decimal, dollar amount, or what

  have you.  There are two main types of variables - numbers and "strings,"

  which are text variables.  Variables are given their type by a symbol

  following their name.

        The "numbers" category is further broken down into four areas.  The

  regular type, called integers, require no symbol, or can be followed by a

  %.  They can be in the range of -32767 to 32767.  Integers are what you

  will be using most of the time.  Another type of integer, long integers,

  have a range of -2 billion to 2 billion.  You might ask, "Why not make

  all numbers long integers?"  There is a simple answer.  The memory of the

  computer, especially in QBasic, is limited - you should save as much space

  as possible.  Only use long integers where they are necessary.  Long int-

  egers are noted with an &.

        The third and fourth types of numbers are "floating-point" numbers.

  These are decimal variables that can have very long decimal spaces.  Again,

  there are both the short types (noted by the !) and long types (noted by a

  #).  These variables are usually not used unless you are doing specific

  accounting functions.

        You still may be in the dark as to how variables are used.  Variables

  are assigned a value using the LET command.  For example:

  LET number = 123

  +-----------------------------------------------------------------------+

  | shortcut:  You don't need the LET in front of the variable assignment |

  +-----------------------------------------------------------------------+  

        This would assign a value of 123 to the short integer variable

  "number."  You can use math functions while assigning variables, too.

  "LET number = 4 * 12" would make "number" equal 48.  You can increment

  variables like this:

  LET number = number + 1

        Or, you can make it "number = number + 2, number = number - 1," and

  so on.  You can also add two variables together using the same syntax, or

  sequence of commands.

        So now you know how to assign a value to a variable using the LET

  command.  You need a way to output these variables to the screen if you

  want them to have any meaning to the user of the program.  You can use

  the PRINT command for this task easily:

  PRINT number

        This would output to the screen the value assigned to "number."

  If you want to include text before the number, you must use this format:

  LET number = 100

  PRINT "The number is"; number

        This would output, when run:

  The number is 100                

        Pretty simple stuff.  You might now be asking, "How do I ask the user

  for something?"  You do this by using the INPUT command.  INPUT basically

  uses the same format as PRINT.  As an example:

  INPUT number

  PRINT "The number is"; number

        This would give the user a prompt for a short integer to assign to

  "number" and then print it out.  You can also supply a prompt for INPUT

  to use - like this:

  INPUT "What is your name"; username$

  PRINT "Hello, "; username$; "."

        This would ask the user for a string and then output it with the

  PRINT command.

        In this chapter, we've gone into a bit more depth with the PRINT

  command and learned about the LET and INPUT commands.  Please wait until

  Chapter III is done - expected date:  6/10/96.

Exercises:

        1.  Fool around with both the LET and INPUT commands.

        2.  Write a program to input the user's name and age and output

        them using the PRINT command.

                              author's note

Sorry it's been so long since I wrote an installment in my book, but I've had

so much going on in the past six months!  However, I finally sat down and

wrote a rather long chapter that explains a LOT of stuff that's really useful.

If you need help with specific commands, be sure to write me!

David Zohrob

mallard@qbasic.com

1/11/1996

        Welcome!  In this chapter, we will cover the following commands:

  þ DIM

  þ FOR...NEXT

  þ STEP

  þ GOTO

  þ IF...THEN

  þ COLOR

        As of right now, you know about variable types, PRINT, INPUT, and

  LET.  These are the most basic parts of the BASIC environment.  We now

  have to move into a more complex area.

        Numbers are used in every application, no matter what kind of program

  it is.  That's why variables are so important in BASIC.  However, when

  dealing with large amounts of numbers and strings, hundreds of variable

  names would get too complex and unreadable.  This is where ARRAYS come in.

        Think of an array as a big box in the computer's memory seperated

  by dividers into smaller boxes.  You can put different things in the small

  boxes, but organize them into one big box.  This is exactly what an array

  does.  An array is an area in the computer's memory seperated into smaller

  parts for different variables.

        To create an array, you use the DIM command:

        DIM ArrayName$ (number)

        ArrayName is the name of the array, followed by a variable type ($,

  !, etc.)  The number in parentheses is the amount of smaller "boxes" in

  the array.

        To use an array, you assign values to the array using the LET command

  just as you would a regular variable.  However, this time you have to spec-

  ify which part of the array you are assigning the value to.  For example,

        DIM Day$(7)

        LET Day$(1) = "Sunday"

        LET Day$(2) = "Monday"

        and so on.  To PRINT one of these variables in the array, you would

   do so just like a regular variable:

        PRINT "Today is "; Day$(1); "."

        And, accordingly, to get a value from the keyboard into the array,

   you use the INPUT command:

        INPUT "What is Player 1's name"; players$(1)

        INPUT "What is Player 2's name"; players$(2)

        You can use an array element just as you would use a variable in

   any statement.  Using arrays, you can manage hundreds of names, values,

   strings, dollar amounts, and much more.

        So now you can use arrays.  But wouldn't it be tedious, say, if you

   wanted to print the 100 different values of an array named players$?  It

   obviously would be wasteful to have 100 different print commands.  This

   is where LOOPS come in.

        A loop is a set of instructions that are repeated either a certain

   number of times, until a condition is met, or infinitely.  The first loop

   command (or set or commands, really) we will study is the FOR...NEXT loop.

   It repeats the commands between the FOR and NEXT commands a certain number

   of times.  Its syntax is simple:

        FOR i = 1 TO 100

        PRINT i

        NEXT i

        The program would print the numbers from one to a hundred.  Why did

   we name the variable "i"?  It's just a standard - one that has been in

   place for several decades.  You can call it anything you want - "bob,"

   "sam," "linda," or "z."  You just have to adjust the code above.  You can

   also change the lower and upper values of i - for example,

        FOR i = 6 TO 371

        PRINT "The magic number could be:"; i

        NEXT i

        Another way to change how the FOR...NEXT loop increments is by using

   the STEP command.  If you omit the STEP command (as in a regular FOR...NEXT

   loop), "i" increases by one every time the commands in the loop are comp-

   leted.  With the use of the STEP command, you can change this to any number

   you desire.  For example:

        FOR i = 100 TO 300 STEP 2

        PRINT "The magic number might be:"; i

        NEXT i

        This would print all the even numbers from 100 to 300.  You could also

   make "i" go backwards, as in this example:

        FOR i = 300 TO 100 STEP -2

        PRINT "The nifty numeral is now:"; i

        NEXT i

        FOR...NEXT loops can be used for just about anything.  In the next

   chapter, I will talk about two other kinds of loops that keep going until

   a certain condition is met.  But until then, (it won't be six months this

   time!) fool around with this loop.

        The last command I'll talk about in this installment is the GOTO

   command.  If you don't know about it, GOTO is a very simple yet extremely

   powerful command that lets you split up your program into smaller parts.

   Here's how you use it.  You have to "label" lines that you want to go to

   using the GOTO command by assigning them numbers or names.  For example:

   topofprogram: CLS

   PRINT "This is an infinite loop"

   GOTO topofprogram

        Or, this:

   1 CLS

   PRINT "Hello world!"

   GOTO 1

        And that's GOTO.  Very simple, yet powerful.  So how can it be use-

   ful?  It's not very important to be able to just GOTO in a program without

   some reason to.  GOTO becomes very useful when used in conjunction with

   the IF...THEN statement, which allows logical progression of your program

   based on a condition.

        IF...THEN does exactly what its commands say.  IF this is true, THEN

   do this.  You can use it to create some type of menu, like this:

   PRINT "My Menu"

   PRINT "Press 1 to clear the screen, or 2 to say 'Hello'!"

   INPUT "What do you want to do"; choice

   IF choice = 1 THEN GOTO clrscr

   IF choice = 2 THEN GOTO hello

   clrscr:  CLS

   PRINT "Done."

   END

   hello:  PRINT "Hello, hello, hello!"

   END

        You can change the GOTO after the THEN.. statement to any valid

   QBasic command, like LET, PRINT, or INPUT.  Or, you can replace the

   "=" with any mathematical symbol (like greater than(>), less than(<>)) Here's a couple of examples in one program:

   PRINT "Program Example #1"

   PRINT "Try to guess the number I am thinking of between 1 and 10."

   PRINT "You get 3 chances."

   INPUT "First chance"; number

   IF number = 3 THEN GOTO gotit

   PRINT "Sorry!"

   INPUT "Second chance"; number

   IF number = 3 THEN GOTO gotit

   PRINT "Ooh!"

   INPUT "Last chance"; number

   IF number = 3 THEN GOTO gotit

   PRINT "Sorry!  The number was 3!"

   END

   gotit: PRINT "You win!  Good job!"

        And that's your basic IF...THEN statement.  The last thing we'll

   deal with in this chapter is a cosmetic command that will let you change

   the color of the text on the screen.  It is the COLOR command, and lets

   you change the color of the text to one of 16 colors (including black,

   the background color).  You use it in this form:

   COLOR 13

   PRINT "Magenta!"

   COLOR 7

   PRINT "Grey!"

        The number after the COLOR statement is one of these color codes:

           00 - black                    08 - dark grey

           01 - dark blue                09 - light blue

           02 - dark green               10 - light green

           03 - dark cyan                11 - light cyan

           04 - dark red                 12 - light red

           05 - dark purple              13 - magenta

           06 - orange brown             14 - yellow

           07 - grey                     15 - bright white

        That ends our discussion of the COLOR statement and our chapter.

   This chapter introduced a lot of new material that will allow you

   to accomplish a lot more in your QBasic programs.  Good luck, and

   keep bugging Mallard to write a fourth installment in this series!

   Exercises:

        1.  Write a guessing-game program that gives you three chances

        to guess a pre-determined number between 1 and 10.  After each

        guess, if the guess is incorrect, have the program tell the user

        whether (s)he is too high or too low.  If the user guesses

        correctly, give them a colorful "winner" message.

        2.  Create an array that stores all the months of the year, then

        create an array with all the days of the year.  Write a program

        that displays all the months and days in a colorful form using

        a FOR...NEXT loop.

        3.  Fool around with all the commands in this chapter.

        4.  Have fun with programming and keep checking The QBasic Page

        for programs, tutorials, and QBasic fun!

                              author's note

Okay, so here's the next chapter.  Sorry it takes a while between chapters

(like two months), but you've all probably heard my stories of the U of D

musical, so..  just enjoy, learn a bit, and give me an e-mail with what you

think about my next installment.

-Dave Zohrob

5/8/1997

        Welcome!  In this chapter, we will cover the following commands:

  þ DO...LOOP                           þ  OPEN

  þ INT                                 þ  CLOSE

  þ RANDOMIZE TIMER                     þ  RND                                 þ  PRINT #

  þ SELECT...END SELECT                 þ  INPUT #

  þ INKEY$

        So you haven't read a tutorial in two months or so.. and now, I'm

  going to satisfy your programming appetite with some juicy commands to

  spice up your programs.  The first command we'll discuss here is the

  DO...LOOP type of loop.  In the previous chapter, we talked about the

  FOR...NEXT loop, which let you do something a certain number of times.

  This is great if you know EXACTLY how many times you want something to

  be done.  But what if you don't?  What if you want something to be done

  until a condition is filled?  This is exactly the void the DO...LOOP

  loop fills.

        All you have to do is say DO WHILE... or UNTIL.. a condition is

  met, such as "x = 10".  Then put the commands you want to be in the loop,

  and have a LOOP command to close it all up.  Here's a very simple example

  of how to use the DO loop.

  DO UNTIL a = 10

  try = try + 1

  PRINT "Try number"; try

  PRINT

  PRINT "Guess the super-secret number!"

  INPUT "Well"; a

  LOOP   

        Of course, there's at least two ways to do everything in programming,

  so instead of DO UNTIL a = 10 as the first line, you could use the line

  DO WHILE a <> 10 and it would do the same thing.  That's it!  The loop

  is very simple, but infinitely useful.

        One such use of the DO loop is to have a "Press Any Key" prompt that

  new programmers are so eager to stick in their programs, but usually have

  to settle for "Please press enter... ".  This is accomplished with

  the nifty function INKEY$, which is the IMMEDIATE, CURRENT key being

  pressed on the keyboard at any given moment.  To make a "Press a key"

  prompt, you can cut and paste this next little bit into your program.

  DO WHILE INKEY$ = ""

  LOOP

        And that's it!  Holy posable action figures, Batman!(tm)  INKEY$ also

  has tons of uses in your program, and one of its main uses is to have

  a "hot-key" menu instead of an INPUT menu.  However, this requires a more

  complex construction similar to an IF...THEN construction.  This new

  set of commands is the SELECT CASE...END SELECT group, and lets you set

  a bunch of different IF...THEN like statements into one big grouping in

  your program instead of having hundreds of seperate statements.  It also

  allows you to have multiple commands per CASE of a variable.

        The use is pretty simple, as shown in this example:

  PRINT "The Main Menu"

  PRINT "1) End the program"

  PRINT "2) Surprise"

  PRINT

  INPUT "Choice"; chc

  SELECT CASE chc

        CASE 1

        PRINT "Fine, then!"

        END

        CASE 2

        PRINT "Surprise! AAAH!!"

        PRINT "Are you surprised?  No?  Oh well..."

        END

        CASE ELSE                                  'note this little

        PRINT "Why didn't you pick 1 or 2?"        'command which lets you

        END                                        'trap invalid answers

  END SELECT

        All you need to do is have a SELECT CASE variable statement, then

  give all the seperate CASEs as you would an IF statement.  When you're all

  done, be SURE to stick an END SELECT in there or else your program won't

  run.

        One other nifty programming trick I stuck in there was the CASE ELSE

  command.  This checks to see that the variable does equal one of your

  CASEs, and if it doesn't, executes the commands under its CASE.  And that's

  the basics of the SELECT CASE..END SELECT group of statements, which you

  will probably see a lot in programs written by myself just because they're

  very useful commands.

        Another very requested item is how to do random numbers in QBasic

  for dice rolls or whatever you might need it for.  There's a very simple

  way to do this, involving the LET, INT, and RND commands.  Here's the syn-

  tax of it:

        x = INT(RND * 10) + 1

        This would give you a number between 1 and 10.  To get a number

  between 0 and 10, jus get rid of the "+ 1" portion of the line.  That's

  all you have to do for random numbers - just change the RND number to get

  your upper bound.

        Now, on to the file commands.  This really should be covered in a

  different chapter, but due to the overwhelming demand for a files tutorial,

  here's the basics of how to use files in QBasic.  The kind of files we'll

  be dealing with are called "sequential files."  There's not much to that

  complex term (why use a big word when a dimunitive one will do?).  All it

  really means is that we write to the file one line at a time, from top to

  bottom.

        You need to use the OPEN command to open a file and the CLOSE command

  to close it (wow! try and follow that logic!).  Be sure to CLOSE every file

  that you open before your program terminates so that your data is saved.

  To open a file, you have to use the command in this way:

        OPEN "filename.ext" FOR [OUTPUT/INPUT/APPEND] AS #1 [or other number]

        The OPEN command is far more complex than this, but this is as far

  as I'll go for now.  The [OUTPUT/INPUT/APPEND] portion changes, depending

  on what you want to do with the file you've opened.  If you want to read

  from the file, line by line, you have to use the INPUT command.  To write

  to the file, overwriting whatever is already there, you have to use the

  OUTPUT command.  And to append to the file (add on to the end), you use

  the APPEND command. See how all that works out?

        So here we are.  We've opened our file.  Now what do we do?  I know

  you still remember the PRINT and INPUT commands (how could you not?), so

  depending on whether you're INPUTting from or OUTPUTting to, there are

  two things to do.  You still use the PRINT command to print and the INPUT

  command to read from the file, but now you stick a file number (like #1)

  in front of what you want to PRINT or INPUT.  For example:

  OPEN "file.txt" FOR OUTPUT AS #1

  PRINT #1, "Hello world!"

  CLOSE

  OPEN "file.txt" FOR INPUT AS #1

  INPUT #1, s$

  PRINT s$

  CLOSE


And that's how you use files.  Pretty simple for sequential files,

  huh?  Good.  You're going to need to fool around with these commands a 

  LOT before you get the hang of it.  There's one more thing to say about

  files, though... the INPUT # statement only reads up to a comma on a

  line.  So, you must use LINE INPUT to read an entire line.  For example:

  OPEN "file.txt" FOR INPUT AS #1

  LINE INPUT #1, s$

  PRINT s$

  CLOSE


This would read the entire line into the variable "s$".  Just re-

  member, when using the OUTPUT command, it ERASES what is in the file al-

  ready.  Be sure not to get rid of anything important!  


So that's the fourth installment in my little series here.  I hope

  you learned a lot, and don't worry, it won't be that long before I put 

  out a fifth chapter.  So keep programming, and try these exercises:

  *exercises*

  1.  Make a guessing game that starts with a random number between 1 and 

  20.  Give the player 5 guesses, and after each guess tell them whether

  they are too high or too low.

  2.  Use the file commands a LOT.  And I mean a LOT.  If you're making a

  game, try to create a high score list or a "save game" feature.

  3.  Keep at it!

                        Basic Basic -  chapter V

                    by David Zohrob (dr.qbasic@qbasic.com)

                            http://qbasic.com

                              [author's note]

Wow!  Almost two months between tutorials TO THE DAY.  Anyway, I'm going to

cover a LOT of ground in this tutorial, so PLEASE pay attention.  Got your

#2 pencil in hand?  Okay.  Mark the ovals with a DARK, COMPLETE circle...

-Dave Zohrob

7/7/1997

        Welcome!  In this chapter, we will cover the following commands

    and topics:  (finally!  I haven't changed that sentence since chapter 2!)

  þ WHILE...WEND                        þ  Random Access Files

  þ APPEND                              þ  GET (file I/O)

  þ SUB...END SUB                       þ  PUT (file I/O)

  þ FUNCTION...END FUNCTION             þ  LEN

  þ GOSUB...RETURN                      þ  TYPE...END TYPE

  þ DIM SHARED                          þ  COMMON SHARED

  þ RTRIM$                             

        Sorry about the delay between tutorials, but I'm going to write one

  RIGHT NOW just for you, a QB-hungry programmer.  I'll also write another

  one this week because of my guilty conscience in not providing enough fodder

  for you guys to learn.  Enough of the "Dave's Thoughts," though, let's get

  right to it.

        Let's get right to the meat of things.  Don't be scared by the amount

  of stuff listed up there, there's a lot of important information here and

  it's not that hard to learn.  We'll cover another method of looping, ways

  to add on to the end of sequential files, structured programming, and

  "random-access" files.  That's a lot of material.

        First off, the last of the looping techniques.  We've already tried

  DO...LOOP and FOR...NEXT, and here comes WHILE...WEND.  It's very similar

  to a DO...UNTIL loop, and I rarely use it in my programs, but you might

  as well learn everything you can.  Its basic syntax is as follows:

  WHILE 

  ...

  WEND

        That's it.  Let's say you want to make a simple guessing game.  All

  it takes is a simple loop, as in the next example.

  RANDOMIZE TIMER

  PRINT "Guessing game!"

  number = INT(RND * 10) + 1                    'random number from 1 to 10

  PRINT "The number is from 1 to 10... go!"

  WHILE guess <> 10                             'loop until the guess = 10

  INPUT "Guess"; guess

  tries = tries + 1                             'count the number of tries

  WEND                                          

  PRINT "Good job!  You got it in"; tries; " tries!"

  END

        There's a working, somewhat interesting program in only 10 lines!

  We're finally getting somewhere.  That's all there is to a WHILE...WEND

  loop.  So simple, and yet so useful!

        We talked about sequential files in Chapter 4.  Those were the files

  that you could only INPUT from or OUTPUT to at one time.  There's one more

  way to add to these files, and that is done by using the APPEND command.

  You use it just like OUTPUT and INPUT:

  OPEN "file.txt" FOR APPEND AS #1

        APPEND is just like OUTPUT, except it doesn't erase what's already

  in the file before it adds to it.  It's useful for ongoing high score lists

  and a bunch of other things that I'm sure your creative mind can think of.

  Again, not much to it.  Just fool around until you get the hang of it.

        Moving on, we'll now discuss structured programming.  Programs are

  huge jumbles of thousands of lines of code, and if everything was just

  arranged from top to bottom, it would take forever to find a problem in

  the program!  Plus, it just looks messy and is hard learn from.  So,

  the geniuses of the 1960s developed something called "structured program-

  ming."  Using this technique, we can create "subprograms" and "functions"

  in our programs to better arrange our code.

        In order to create and edit a subprogram in QBasic, all you have to

  do is go to the [E]dit menu on the menu toolbar and choose New [S]UB...

  QBasic will then prompt you for its name.  Don't get too creative and call

  it "Bob" or "Jane," give it a name like "DoGraphics" so you can tell what

  the SUB really does.

        QBasic will then let you edit the SUB.  Program just as you would

  normally, and when you're done, go to [V]iew on the toolbar, choose [S]UBs

  and then pick your program's filename off of the menu.  If you want to go

  back and edit your SUB, go to the [V]iew menu, choose [S]UBs and pick your

  SUB off the list.  You can have as many subroutines as you want in a pro-

  gram.  Be aware, though, that variables you create in a SUB only are

  accessible in the SUB and variables in the main body of the program are

  only accessible from the main body of the program unless you make them

  SHARED using the following command:

  COMMON SHARED variable$, variable2, variable3!

        This statement goes after your "DECLARE SUB..." declarations (QBasic

  places these statements in your program once you save it) and before any

  other executable lines of your program.  To make an array SHARED, you just

  stick the word SHARED in after the DIM command, like this:

  DIM SHARED array$(100)

        Make sure that any variables you want to be SHARED (called "global"

  variables) are in your COMMON SHARED statement.  Otherwise, your program

  will turn up a lot of errors.  To pass variables on to the SUB as you call

  it from the main program, you must change the name of the SUB.  While you

  are editing it, just use a line like this:

  SUB DoBox (x1, y1, x2, y2)

        QBasic sticks the "SUB DoBox" in there for you.  Now, when you call

  the SUB from the main program, use a line like this:

  DoBox 30, 20, 50, 20

        Otherwise, you can just type the name of the SUB in the main program

  to call it up, like this:

  mainloop: CLS

  DoGraphics

  DoLevel

  WaitForKey

        As a complement to SUBs, those programming gurus of the '60s also

  developed FUNCTIONS.  These are similar to SUBs, but they basically just

  perfom a specific calculation to return a number.  You create them the

  same way you would create a SUB, by going to the [E]dit menu, but choose

  New [F]UNCTION.  You can switch what you're editing the same way.  Here's

  a short example:

  FUNCTION Cube(num)

        Cube = num * num * num

  END FUNCTION

  [main program:]

  CLS

  INPUT "Number"; number

  num3 = Cube(number)

  PRINT number; "cubed ="; num3

  END

        The last topic I'll cover in this tutorial is random-access files.

  These files are kind of like database files - they have a set record type

  and a number of records.  They're very useful for database applications.

  To make the record, you have to define it near the top of your program

  with a TYPE...END TYPE clause.  Here's a short example for an address

  program:

  TYPE people

        nm AS STRING * 40            'set the name as a 40-character string

        age AS INTEGER               'set the age as an integer

        address AS STRING * 60       'set address as a 60-character string

  END TYPE

        The next thing you have to do before you open the file is use a

  DIM command to set a random-access array up for the TYPE, as follows:

  DIM person AS people

        Now, you must open the file.  You again use the OPEN command, but

  now you have to add in two commands: LEN and RANDOM.  Here's an example:

  OPEN "address.dat" FOR RANDOM AS #1 LEN = LEN(person)

        This will open up your random-access file of addresses with the

  record length that you set in your TYPE.  Now, in order to get records

  or put records to a file, you have to use special variable names and

  different file commands.  Here's a quick example of adding a record to

  the file we created above:

  INPUT "What record to add"; record

  INPUT "Name"; person.nm

  INPUT "Age"; person.age

  INPUT "Address"; person.address

  PUT 1, record, person

        As you can see you have to have your DIMmed array name, then a

  period, then the variable name you set in your TYPE.  Then, you have to

  PUT the record to the file using the PUT command (how appropriate).  The

  syntax for the PUT command is like this:

  PUT [filename], [recordnumber], [arrayname]

        It's very simple.  To get an array from the file, you use basically

  the same method, except with the GET command. The GET command has exactly

  the same syntax as the PUT command, except it reads from the file into

  the array that you specify.  Here's an example from the program we have

  above.

  INPUT "View which record"; record

  GET 1, record, person

  PRINT "Name"; person.nm

  PRINT "Age"; person.age

  PRINT "Address"; person.address

        That's how you use random-access files.  An example is available,

  called RANDOM.BAS.  If you didn't download it with the .ZIP file, please

  download it from http://www.qbasic.com/random.bas.  Random-access files

  are very useful for a lot of applications, but they're also very complex.

  One last thing before I conclude - when an input doesn't fill up all of

  the STRING you specified in the array (like in the above example: if

  the person's name isn't 40 characters) the variable WILL have all the

  extra spaces on the end to fill up the 40 characters.  To remedy this,

  use the RTRIM$ command, as follows:

  personname$ = RTRIM$(person.nm)

        This will trim all the extra spaces off of the end of the string.

        And that's it for this tutorial.  There's a heck of a lot of stuff

  here, and if it's confusing, please ask me questions.  They might take a

  long time to answer, but I need to know how to improve my tutorials so

  they help YOU the most.

  Exercise!

        1.  Create a simple database program for addresses and names

            using structured programming.

